

# RISK EVALUATION AFTER HEART VALVE REPLACEMENT BY SAS PROC PHREG

Chao L. Chen and Lynn Wang

Deborah Heart and Lung Center, Browns Mills, NJ 08015

## SUMMARY

Identification of long-term risk factors after heart valve replacement can be quite challenging due to the nature of the covariates. Two specific factors need to be considered: (1) time-dependent covariates, and (2) the validity of the proportional hazard model. PHREG in SAS® is a very convenient procedure for solving these problems. In addition to those commonly encountered problems such as variable selection and influence assessment, we will use a data set from our St. Jude Medical valve prosthesis database to illustrate how to use DATA step statement to incorporate time-dependent covariates into the proportional hazard model and how to plot martingale residuals to check model validity. Results from our analysis indicated that cardio-pulmonary bypass time during surgery should be treated as time-dependent because its adverse effect faded away after six months. The proportional hazard model fitted this data set reasonably well once the effect of time-dependence was taken into consideration.

*Key words:* time-dependent covariate, proportional hazard, martingale residuals

## 1. Introduction

Conventional approaches to survival analysis involve computation of Kaplan-Meier estimates, evaluating the effect of time-independent covariates, and parametric fitting of survival time. SAS® PROC LIFETEST and PROC LIFEREG sufficiently cover the need in this area. These approaches can be inadequate in certain situations. For example, time-dependent covariates and recurrent events may arise in practical applications. Researchers may also be cautious to the liberal assumption of a parametric failure time distribution.

A general approach to deal with these problems is through the theory of counting process (Fleming and Harrington, 1990). Recently, there has been significant advance in software development along this line of thought (Harrell, 1997) and SAS PROC PHREG is one of them. Though a powerful and versatile tool, the complexity of the underlying mathematical theory is a barrier to general users. To promote the appropriate use of this powerful tool with confidence, we provide numerical examples to elucidate the analogy between procedures derived from counting process theory and regression analysis.

## 2. The data set and the goal of data analysis

From May 1982 to August 1991, 1200 consecutive patients received St. Jude Medical valve prosthesis at Deborah Heart and Lung Center. Due to the diversity of surgical procedures, only hospital survivors with isolated aortic valve replacements were considered in the current study. Even though certain risk factors such as age, preoperative New York Heart Class, cardio-pulmonary bypass time during surgery, postoperative valve gradient, postoperative complications, and so forth, are well known to the clinicians, lack of a numerical model makes prediction of prognosis difficult. Our goal is to quantitatively evaluate covariates as risk factors for late death in a numerical model. A complete list of covariates and their acronyms used in our SAS statements are listed in the Appendix. A detailed explanation of certain clinical terms can be found in Young et al. (1988).

The adoption of PROC PHREG to construct a quantitative model was prompted by the necessity to treat certain covariates as time-dependent. Certainly, the adverse effect of post operative complications happens only after, not before the complications. In addition, we conjecture that the adverse effect of longer bypass time is most significant immediately after surgery and then declines. This conjecture was confirmed in our analysis.

Another factor that deserves special consideration is postoperative valve gradient. Previous studies indicated that the gradient obtained from echocardiogram is an important indicator for the long-term recovery after surgery (DUMESNIL and YOGANATHAN, 1992, FERNANDEZ, et al., 1994). However, for our aortic group, only 44.2 % (250/565) of the patients had a postoperative echocardiogram. Patients without a postoperative echocardiogram were not included in our analysis. The mechanisms of missing value and its consequences will not be considered in this study.

## 3. Proportional Hazard model with time-independent covariates

### 3.1 Variable selection

SAS PHREG procedure performs a regression-type analysis based on a model proposed by Cox (1972). It assumes that  $\lambda(t/Z)$ , the hazard of patients at time  $t$  with covariates  $Z$ , is equal to  $\lambda_0(t) \cdot \exp(\beta'Z)$  where  $\lambda_0(t)$  is the baseline hazard and  $\beta'$  are the coefficients to be estimated. In our study, the following SAS statements

were used to evaluate the significance of these coefficients.

Output 1.

Step 5: Variable EMVGR is removed. The model contains the following explanatory variables.

PRENYH EAVGR PUMP

Analysis of Maximum Likelihood Estimates

Variable	DF	Parameter Estimate	Standard Error	Wald Chi-Square	Pr > Chi-Square
PRENYH	1	0.922365	0.38885	5.62654	0.0177
EAVGR	1	0.067521	0.02421	7.78077	0.0053
PUMP	1	0.009207	0.00578	2.53619	0.1113

```
proc phreg data=b;
model folint*flag(1) = age prevs prenyh
mi eavgr emvgr vai pump/
selection=backward stop=3;
run;
```

In this example, the SAS dataset to be analyzed is stored in b. We tried to include an exhaustive list of covariates related to long-term survival rate and opted the backward elimination with minimal number of covariates no fewer than three. This is achieved by using the optional statement “selection=backward stop=3;”.

Other approaches to variable selection such as forward and stepwise selection are also available, see SAS manual (1996) for details. This convenient tool, not always available in other software packages, makes SAS PHREG an ideal tool for initial screening for important covariates to be included in the proportional hazard model. Partial results were listed in Output 1. Three variables, “prenyh”, “eavgr”, “pump”, were left in the final step of the backward elimination (Step 5 in Output 1). These three variables were used to illustrate how model checking can be performed by plotting the associated residuals.

### 3.2. Model checking

Parameter estimation and test of significance are by no means the end of statistical analysis. It is a common practice in regression analysis to check the model adequacy by plotting residuals and to evaluate influential observations. Similar procedures can be applied to Cox proportional hazard model, too. Plotting of martingale residuals can be used to evaluate whether a specific time independent covariate can be entered directly into the model, or a transformation is necessary.

For the identification of influential observations, the approximation of change in parameter estimates when each individual observation was deleted can be conveniently performed by an option in SAS PHREG.

#### 3.2.1 Plot of martingale residuals

Martingale residual for observation  $i$  is defined as

$$\hat{M}_i = \delta_i - \int_0^{t_i} \exp(\hat{\beta}' Z_i) d\hat{\Lambda}_0(s)$$

where

$$\delta_i = \begin{cases} 0 & \text{if } t_i \text{ is a censored time} \\ 1 & \text{if } t_i \text{ is an event time} \end{cases}$$

and  $\hat{\Lambda}_0(s)$  is the estimated cumulative hazard rate for the baseline population. When all the elements in the covariate vector  $Z_i$  are time independent, the martingale residuals can be simplified as

$$\hat{M}_i = \delta_i - \exp(\beta_i' Z_i) \hat{\Lambda}_0(t_i).$$

Similar to residuals in regression analysis, martingale residuals can be conceptually explained as the difference of the observed outcome ( $\delta_i$ ) and the estimated cumulative hazard of the  $i$ -th patient ( $\exp(\beta_i' Z_i) \hat{\Lambda}_0(t_i)$ ). It also has some desirable properties such as zero sum and zero covariance. For a detailed discussion, see Fleming and Harrington (1991). A useful result there states that, if the true hazard function is  $\exp(f(X) + \beta'Z) \cdot \lambda_0(t)$ , then given the covariate  $X$ , the conditional expectation of martingale

residuals computed from the model without the covariate  $X$  is approximately equal to  $a \cdot f(X) + b$ , where  $a$  and  $b$  are constants. Therefore, a plot of martingale residuals vs. the covariate  $X$ , together with a smoothed curve estimate provides clues about the appropriate functional form of  $X$  to be entered in the proportional hazard model. Straight line indicates no transformation is necessary.

The corresponding SAS statements listed below exemplified using this technique on our dataset. In the model statement, “prenyh” is omitted and then the martingale residuals were plotted against “prenyh”. Similar procedures can be applied to the plot of “eavgr” and “pump”. Note that “resmart=mart” in the output statement specifies the martingale residuals, the optional statement “order=data” preserves the original order so each martingale residual can be combined with its associated observation.

```
proc phreg data=b;
model folint*flag(1) = eavgr pump;
output out=outp resmart=mart
resdev=dev/order=data;
run;
data m; merge outp b;
proc plot; plot mart*prenyh="*";
run;
```

In Figures 1-a and 1-b, the S-PLUS® function loess.smooth was used to generate a smoothed line. This enables us to visually check the trend. Only the results for “eavgr” and “pump” were illustrated here, but all the three covariates, including “prenyh”, showed the same linear trend. Therefore, no transformation or dichotomization is necessary in the construction of our proportional hazard model

### 3.2.2 Identification of influential observations

As discussed in the previous section, martingale and other forms of residuals provide us with clues for covariate transformation, these residuals also screen out observations which are not well fitted by the model. However, just as in regression analysis, observations with good fit might also bear undue influence, in other words, inclusion or exclusion of those highly influential points might drastically change the estimates. In regression analysis, the impact of these influential points can be reduced by certain robust regression procedures (HOLLAND and WELSCH). As a first step to evaluate the impact of the influential observations, an option in PHREG, “dfbeta” can be used. The SAS codes for our example were listed below.

```
proc phreg data=b;
model folint*flag(1) = prenyh eavgr
optotpum;
output out=outp dfbeta = dprenyh deavgr
doptotp/order=data;
data m; merge outp b;
i+1;
proc plot data=m; plot (dprenyh deavgr
doptotp) * i ="*";
run;
```

Figure 1-a: Plot of Martingale residual vs. “eavgr”.





Figure 1-b: Plot of Martingale residual vs. “pump”.



In the output statement, three newly generated variables, “dprenyh”, “deavgr” and “doptotp”

after “dfbeta= “ are approximate values of the parameter changes for “prenyh”, “eavgr” and “pump” individually. The results were also shown in Figures 2-a, 2-b, and 2-c. For “prenyh” and “eavgr”, the deletion of any observation may create substantial difference in estimate, but it is never greater the one half of the standard error as listed in Output 1. However, for “pump”, the most significant influential point occurred between observations 50 and 100. The approximated change is between 0.004 and 0.006, which is approximately equal to one standard deviation of the estimate as shown in Output 1. This aberrant point is a patient with “pump” equal to 281 minutes, which happens to be the longest bypass time in the whole series of patients. To obtain the exact result, parameter estimates of proportional hazard model were recomputed without this patient. The recomputed value is 0.003. Compared with 0.009 as in Output 1, the change is quite drastic. To put this in another way, for an increase of ten minutes in bypass time, depending on which parameter to use, the hazard can increase either  $\exp(10 \cdot 0.009) \approx 9\%$  or  $\exp(10 \cdot 0.003) \approx 3\%$ . With the deletion of only one observation, this change in estimate is huge and hence the stability of the parameter estimation deserves close attention.

Figure 2-b: Approximated parameter change for “eavgr”.




Figure 2-a: Approximated parameter change for “prenyh”.





Figure 2-c: Approximated parameter change for “pump”.



## Output 2.

Step 5: Variable EMVGR is removed. The model contains the following explanatory variables.

PRENYH EAVGR ADJPUMP

### Analysis of Maximum Likelihood Estimates

Variable	DF	Parameter Estimate	Standard Error	Wald Chi-Square	Pr > Chi-Square
PRENYH	1	0.972087	0.39049	6.19704	0.0128
EAVGR	1	0.071145	0.02318	9.42224	0.0021
ADJPUMP	1	0.031308	0.00734	18.20294	0.0001

4. Proportional Hazard model with time-dependent covariates.

#### 4.1 Pump time as a time-dependent variable

We showed in the previous section that from a statistical viewpoint, estimate of hazard from pump data is unstable. From a clinical viewpoint, there are reasons to support that negative postoperative impact of pump time is limited to the initial period after surgery. To take this perspective into consideration, we can define “pump” as a time-dependent variable as specified in the following SAS statements.

```
proc phreg data=b;
model folint*flag(1) = age nprevs
prenyh mi eavgr emvgr vai adjpump/
selection=backward stop=3;
if folint gt 6 then adjpump=0;
else adjpump=pump;
run;
```

The statement “if folint gt 6 then adjpump=0; else adjpump=pump;” created a new covariate “adjpump” (adjusted pump time) which is equal to “pump” when the follow-up time is within 6 months, and equal to zero otherwise. In other words, the effect of “pump” will be null after 6 months. The interval 6 months was picked because it generated the most significant results among several values tried. The results were listed in Output 2.

Again, the same three covariates come out in the variable selection process. Not only the adjusted pump time “adjpump” generated a more significant p-value, it also greatly increased the parameter estimate from 0.009 to 0.031. This result indicated the negative impact of longer pump time is larger than originally expected, but mainly limited to the first six months after surgery.

One of the limitations in using a time-dependent covariate is that output statement is not allowed, so output data can not be generated. Therefore, model checking procedures previously exemplified in time-independent model can not be applied. As a remedy, we reanalyzed the data with the outlying observation mentioned in the previous session deleted, the reduction of estimate is again about the same magnitude of the standard error of the estimate (0.007).

#### 4.2 Complication as a time-dependent covariate

So far all the covariates in the model are values collected before or immediately after surgery. The versatility of PHREG allows us to incorporate different forms of time-dependent variables. This is achieved by applying SAS statements used in DATA step. As an illustration, we consider adding complications as a time-dependent variable. The SAS statements were listed below.

```
data all; merge aortic aocomp; by mrno;
run;
proc phreg data=all;
model folint*flag(1) =
age nprevs prenyh mi eavgr emvgr vai
adjpump allcomp
/ selection=s sle=0.9;
array event(6) event1-event6;
array evend(6) evend1-evend6;

do i=1 to 6;
if evend(i) eq . then
evend(i)=folldate+1;
compfree=(evend(i)-dos)/30.5;
if ((folint-compfree) gt 0 /*and
(folint-compfree) le 2*/)
then comp=1; else comp=0;
allcomp+comp;
end;
```

```
end;
if folint gt 6 then adjpump=0; else
adjpump=optotpum;
run;
```

The do loop statement created a step function with a jump of one whenever a complication occurred during the follow-up period. The results were summarized in Output 3.

As shown in Output 3, complications also play an important role in long-term survival rate. Notice that we treated “allcomp” as a non-decreasing step function. One may also suspect that, just like “pump”, the adverse effects of these complications may be transient. To take this viewpoint into consideration, we can activate “and (folint-compfree) le 2” in “if ((folint-compfree) gt 0 /\*and (folint-compfree) le 2\*/)”. In this case, we assume that the adverse effect of complications is limited to two months after that specific complication. After trying several different values, we found that assuming a non-decreasing step function has the best fit. In other words, once a patient encountered certain complications, the hazard will never decrease again.

### 5. Discussions and conclusions

In this study, we demonstrated the similarities in model checking between a conventional regression analysis and proportional hazard model. Although the underlying theory of the counting process used in developing these procedures for the proportional hazard model requires much more sophisticated mathematics, the similarities between the two help non-mathematicians

Output 3.

Step 6: Variable EMVGR is removed. The model contains the following explanatory variables.					
PRENYH EAVGR ADJPUMP ALLCOMP					
Analysis of Maximum Likelihood Estimates					
Variable	DF	Parameter Estimate	Standard Error	Wald Chi-Square	Pr > Chi-Square
PRENYH	1	0.932074	0.39690	5.51485	0.0189
EAVGR	1	0.072601	0.02199	10.90321	0.0010
ADJPUMP	1	0.031424	0.00741	17.97086	0.0001
ALLCOMP	1	0.461020	0.19209	5.76033	0.0164

grasp the idea of model checking for proportional hazard model.

When these techniques were applied to our dataset, we found that the most critical covariate for the long-term survival rates are preoperative New York Heart Class (“prenyh”), postoperative valve gradient (“eavgr”), and adjusted cardiopulmonary bypass time (“adjpump”). Two time-dependent covariates, “adjpump” and “allcomp” were considered in the model. Even though both showed significance, we would like to point out that for prediction of prognosis, “eavgr” should be the first covariate to be considered. Late-phase complication information is not always immediately available after surgery, so their prediction value is not as important as other variables considered. Our analysis indicated that once a patient experience certain complications, their hazard will not decrease.

SAS PHREG is a very handy tool to identify the important variables that are critical to long-term survival rate, model checking procedures can be easily performed when no time-dependent variables are involved. Statements used in DATA step can also be implemented to accommodate time-dependent covariates. The covariate “allcomp” was used here as an example to show the versatility of the DATA step in creating different forms of time-dependent variables. The limitation with time-dependent covariates model is that output statement necessary for model checking can not be generated. This somehow limits our capability to perform model checking in a time-dependent model. In addition, survival functions for the baseline population can not be generated when there are time-dependent variables in the model.

Appendix. List of covariates considered in the initial proportional hazard model.

folint : follow-up interval

age : age at operation  
prevs : history of previous cardiac operation  
prenyh : preoperative New York Heart Class  
mi : history of myocardial infarction  
eavgr : postoperative aortic valve gradient

emvgr : postoperative mitral valve gradient  
vai : valve index which defined as prosthetic valve  
area divided by body surface area  
pump : cardiopulmonary bypass time

#### REFERENCES

- Cox, D. R., 1972: Regression models and life-tables (with discussion). *Journal of the Royal Statistical Society, Series B*, **34**, 187-220.
- Flemington, T. R. and Harrington, D. P., 1991: *Counting Process and Survival Analysis*. John Wiley and Sons, New York.
- DUMESNIL, J. G. and YOGANATHAN, A. P., 1992: Valve prosthesis hemodynamics and the problem of high transprosthetic pressure gradients. *European Journal of Cardio-thoracic Surgery*. **6** [Suppl 1]:S34-S38.
- FERNANDEZ, J., LAUB, G. W., ADKINS, M. S., ANDERSON, W. A., CHEN, C., BAILEY, B. M., NEALON, L. M., and MCGRATH, L. B., 1994: Early and late-phase events after valve replacement with the St. Jude Medical prosthesis in 1200 patients. *Journal of Thoracic and Cardiovascular Surgery*. **107**, 394-407.
- HARRELL, JR., F. E. and GOLDSTEIN, R., 1997: A survey of microcomputer survival analysis software: The need for an integrated framework. *The American Statistician*. **51**, 360 - 373.
- HOLLAND, P. W. and WELSCH, R. E., 1977: Robust Regression Using Iteratively Reweighted Least Squares. *Communications in Statistics: Theory and Methods*. **6**, 813-827.
- SAS/STAT software Changes and Enhancements through Release 6.11, 1996: SAS institute, Cary, NC.
- YANG, S. S., BENTIVOGLIO, L. G., MARANHÃO, V., and GOLDBERG, H., 1988: *From Cardiac Catheterization Data to Hemodynamic Parameters*. F. A. Davis Company, Philadelphia.

SAS® is a registered trademark of the SAS Institute, Inc.

S-PLUS® is a registered trademark of MathSoft, Inc.